```
1000REM++
1010REM *HDFORM--Fast winchester disk formatter (DTC-510A controller)
1020REM   Created: 12-NOV-1986   Author: Michael J Tubby   Version 01.00A
1030REM
1040REM NOTE
1050REM
1060REM   Copyright (C) 1986 by Michael J Tubby.  This program may be copied
1070REM   and used providing it is not sold or used for commercial gain.
1080REM
1090REM ABSTRACT
1100REM
1110REM   This program formats a Winchester Disk on the DTC-510A controller.
1120REM   It uses the "format entire media" command to perform the format
1130REM   operation - no verification is done during formatting, any bad
 .40REM   blocks on the media ARE NOT CHECKED/DETECTED - if any are found
1150REM   they will have to be delt with manually (see text).
1160REM
1170REM   After formatting is complete a diagnostic read of the disk is
1180REM   performed, this DOES NOT verify the whole media, it just checks
1190REM   that sector 0 on every track exists and then reads 256 sectors
1200REM   randomly - it checks that the drive is okay...
1210REM
1220REM--
1230DIM buffer% 1000                    :REM dummy buffer space
1240DIM block% 16                       :REM OSWORD buffer
1250
1260sk%=4                               :REM optimum sector skew for disk
1270PRINT"Formatting..."
1280
1290?block%=0                           :REM set up OSWORD parameter block
1300block%?1=buffer%
1310block%?5=4                          :REM format command is 04
1320block%?6=0
1330block%?7=0
 340block%?8=0
-350block%?9=sk%                        :REM sector skew
1360block%?10=0
1370block%!11=0
1380
1390A%=&72:X%=block%MOD256:Y%=block%DIV256
1400CALL &FFF1
1410IF ?block% <> 0 THEN PRINT"Format error":END
1420
1430PRINT"Format complete"
1440
1450PRINT'"Diagnostic read"             :REM do diagnostic read
1460block%?5=&E3
1470block%?6=0
1480block%?7=0
1490block%?8=0
1500block%?9=0
1510block%?9=0
1520
```

```
1530A%=&72:X%=block%MOD256:Y%=block%DIV256:CALL &FFF1
1540IF?block% <> 0 THEN PRINT"Diagnostic read error":END
1550PRINT"Diagnostic completed"
```

```
1000REM++
1010REM *HDINIT--Initialise winchester disk (for DTC-510A controller)
1020REM   Created: 12-NOV-1987   Author: Many       Version: 01.21A
1030REM
1040REM
1050REM NOTE
1060REM
1070REM   This program is copyright.  It may be copied and used as long as
1080REM   it is not sold or used for commercial gain.
1090REM
1100REM
1110REM ABSTRACT
1120REM
1130REM   This program writes the free space map to the winchester and the
1140REM   root directory.  The size of the winchester is determined by values
1150REM   placed in variables at the start of the Init procedure, these
1160REM   should be altered to suite your drive characteristics.
1170REM
1180REM   This program is run after the winchester is formatted to make it
1190REM   useable by ADFS.  If this program is run on a disk that already
1200REM   has data on it it will be lost - be careful!
1210REM
1220REM   This program has it's origins in Acorn's SuperForm (Winchester
1230REM   formatter program for Adaptec controller) and has been modified by
1240REM   serveral people.
1250REM
1260REM   Four "tracks" are reserved at the end of the disk by lowering the
1270REM   maximum available DA to 4*(number of sectors per track).  These
1280REM   tracks are for use as altrenate tracks, they can be mapped in if a
1290REM   sector (block) becomes bad to keep the disk logically perfect (see
1300REM   text).
1310REM
1320REM--
1330PROCinit
1340PROCwrite_init
1350END
1360:
1370DEF PROCinit
1380:
1390@%=4
1400:
1410heads%=4                    :REM number of heads (surfaces) in the drive
1420cyl%=310                    :REM number of cylinders
1430step%=1                      :REM step speed code (not used)
1440rwcc%=128                   :REM reduced write current cylinder (not used)
1450sect%=33                    :REM sectors/track (DTC & Xebec=33, Adaptec=32)
1460:
14701z%=0
1480drive%=0
1490:
1500DIM X% 20
1510Y%=X% DIV 256
1520:
```

```
1530DIM data% 1044
1540!data%=0
1550:
1560DIM buffer% 1280
1570buffer0%=buffer%
1580buffer1%=buffer%+1
1590buffer2%=buffer%+2
1600buffer3%=buffer%+3
1610buffer4%=buffer%+4
1620buffer5%=buffer%+5
1630buffer6%=buffer%+6
1640buffer7%=buffer%+7
1650:
1660:
1670ENDPROC
1680DEFPROCwrite_init
1690FORI%=buffer% TO buffer%+1279 STEP4
1700!I%=0
1710NEXT
1720:
1730!buffer%=7
1740buffer%!252=heads%*cyl%*sect%-(4*sect%)
1750buffer%!256=heads%*cyl%*sect%-(4*sect%)-7
1760buffer%!507=RND
1770buffer%!509=&300
1780:
1790P%=buffer%+1000
1800[ OPT2
1810CLC:LDY£&FF:TYA
1820.L1% ADC buffer%+255,Y
1830DEY:BNE L1%
1840TAX:DEY:CLC:TYA
1850.L2% ADC buffer%-1,Y
1860DEY:BNE L2%
1870STX buffer%+511
1880STA buffer%+255
1890RTS
1900]
1910CALL buffer%+1000
1920:
1930PRINT"Writing map"
1940PROCwrite_map
1950:
1960FORI%=buffer% TO buffer%+1279 STEP4
1970!I%=0
1980NEXT
1990buffer%!1=&6F677548
2000buffer%!&4FB=&6F677548
2010$(buffer%+&4CC)="$"
2020$(buffer%+&4D9)="$"
2030buffer%?&4D6=2
2040PROCwrite_dir
2050PRINT"Writing root"
2060ENDPROC
```

```
2070:
2080DEF PROCwrite_map
2090?X%=0
2100X%!1=buffer%
2110X%?5=10
2120X%?6=drive%*&20
2130X%!7=0
2140X%!9=2
2150A%=&72
2160CALL &FFF1
2170IF ?X% PRINT"Error "~?X%" in write_map":STOP
2180ENDPROC
2190:
2200DEF PROCwrite_dir
2210?X%=0
2220X%!1=buffer%
2230X%?5=10
2240X%?6=drive%*&20
2250X%!7=&50200
2260A%=&72
2270CALL &FFF1
2280IF ?X% PRINT"Error "~?X%" in write_dir":STOP
2290ENDPROC
2300:
```

```
1000REM++
1010REM *HDVERIFY--Hard Disk Verifier (for the DTC-510A controller)
1020REM
1030REM   Created: 12-DEC-1986   Author: Michael J Tubby   Version: 01.00A
1040REM
1050REM
1060REM NOTE
1070REM
1080REM   Copyright (C) 1986 by Michael J Tubby G8TIC.  This program may
1090REM   be copied and used as long as it is not sold or used for
1100REM   commercial gain.
1110REM
1120REM
1130REM ABSTRACT
1140REM
1150REM   This program verifies a hard disk (Winchester) running on a Beeb
1160REM   with the DTC-510A/B controller.  The program checks that the track
1170REM   has a valid format on it (no. of sectors, skew, id field crc etc.)
1180REM   with the "check valid track" command, and then checks the data on
1190REM   the track by reading it into a track buffer.  Unfortunately this has
1200REM   to be done as the old DTC controller does not have the advanced op-
1210REM   codes of the newer ADAPTEC/XEBEC controllers.
1220REM
1230REM   As the controller also does not support the "sense drive
1240REM   characteristics" command, the size attributes of the drive
1250REM   cannot be obtained from the comtroller, so they have to be
1260REM   'hard coded' into the program.
1270REM
1280REM            cyl%            is the number of cylinders on the drive
1290REM            head%           is the number of heads on the drive
1300REM            interleave%     is the sector skew (interleave) value
1310REM
1320REM--
1330
1340DIM block% 16
1350DIM error% 16
1360DIM buffer% 33*256
1370
1380cyl%=310
1390head%=4
1400interleave%=4
1410
1420hs%=33*head%
1430la%=(cyl%-1)*head%*33 + (head%-1)*33 + 32
1440PRINT'"HDVERIFY - Hard Disk Verifier (DTC-510A controller) - Version 1.00"
1450PRINT"Copyright (C) 1986 by Michael J Tubby G8TIC."'
1460INPUT"Verify: Data/Track format/Both ? (D/T/B) :"A$
1470IFA$>"Z"THENA$=CHR$(ASC(A$)-32)
1480IFA$<>"D"ANDA$<>"T"ANDA$<>"B"THEN1460
1490@%=1:PRINT
1500ONERRORGOTO1620
1510
1520da%=0
```

```
1530cy%=da%DIVhs%
1540hd%=(da%MODhs%)DIV33
1550PRINTCHR$13;"Cylinder "RIGHT$("    "+STR$cy%,3)", Head "hd%;
1560IF(A$="T" OR A$="B") THEN IFNOTFNverify(da%,interleave%) THEN z$="Format er
ror ":GOTO1640
1570IF(A$="D" OR A$="B") THEN IFNOTFNread(da%) THEN z$="Data error ":GOTO1640
1580da%=da%+33:IFda%<la%THEN1530
1590PRINT''"Finished.":END
1600
1610
1620IFERR=17THENPRINT''"Escape"':END
1630z$="Disk (ADFS) error "
1640REM
1650er%=FNgeterr
1660error_code%=error%?3 AND &7F
1670PRINT", Sector "er%MOD33 " (DA " RIGHT$("00000000"+STR$~(er%),6) ") "z$;err
_code%
1680ONERRORGOTO1620
1690GOTO1580
1700
1710DEFFNverify(lad%,skew%)
1720?block%=0
1730block%!1=buffer%
1740block%?5=5
1750block%?6=lad%DIV &10000
1760block%?7=lad%DIV &100
1770block%?8=lad%MOD256
1780block%?9=skew%
1790block%?10=0
1800block%!11=0
1810
1820X%=block%MOD256:Y%=block%DIV256:A%=&72:CALL &FFF1
1830IF?block%=0 THEN=TRUE ELSE=FALSE
1840
1850DEFFNread(lad%) :REM read 33 sectors of a (whole) track to buffer
1860?block%=0
1870block%!1=buffer%
1880block%?5=8
1890block%?6=lad%DIV &10000
1900block%?7=lad%DIV &100
1910block%?8=lad%MOD256
1920block%?9=33
1930block%?10=0
1940block%!11=0
1950X%=block%MOD256:Y%=block%DIV256:A%=&72:CALL &FFF1
1960IF?block%=0 THEN=TRUE ELSE=FALSE
1970
1980DEFFNgeterr :REM returns the DA of the error
1990X%=error%MOD256
2000Y%=error%DIV256
2010A%=&73:CALL &FFF1
2020=!error% AND &FFFFFF
```